Positivity Proof on an Example

June 2025

Aim of this example: prove that the sequence defined by

```
> rec:=(16*n+1)*u(n+3)-(32*n-2)*u(n+2)+(20*n-4)*u(n+1)-(5*n-3)*u(n)=0;
```

$$rec := (16\,n+1)\,u(n+3) - (32\,n-2)\,u(n+2) + (20\,n-4)\,u(n+1) - (5\,n-3)\,u(n) = 0 \quad (1)$$

and

```
> ini:={u(0)=5,u(1)=5,u(2)=1};
```

$$ini := \{u(0)=5, u(1)=5, u(2)=1\} \quad (2)$$

is positive.

```
> with(LinearAlgebra):
```

Check the first values

```
> N:=20:
> L:=gfun:-rectoproc({rec,op(ini)},u(n),list)(N);
```

$$L := \left[5, 5, 1, 3, \frac{84}{17}, \frac{3491}{561}, \frac{193118}{27489}, \frac{389756}{51051}, \frac{79660582}{9648639}, \frac{765516302}{85083453}, \frac{1047161107676}{105758732079}, \frac{116430530626}{10600525593}, \frac{734444332400872}{59945972228415}, \frac{1456409614649458996}{106164316816522965}, \frac{4134666999908131424}{268444058236065211 5}, \frac{6284662466466527841063 2}{3626679226769241007365}, \frac{148031580601376890234887 76}{757975958394771370539285}, \frac{2506108194731086273364863 52}{113696393759215705580892 75}, \frac{682516900404428312453981658 56}{274008308959709850449951527 5}, \frac{19843[...22\ digits...]07128}{704201354026454315656375425675}, \frac{18588[...23\ digits...]31504}{58256[...21\ digits...]39675}\right] \quad (1.1)$$

```
> evalf(%);
```

[5., 5., 1., 3., 4.941176471, 6.222816399, 7.025282840, 7.634639870, 8.256147007, 8.997240650, 9.901415109, 10.98346772, 12.25177114, 13.71844758, 15.40234128, 17.32897252, 19.52985170, 22.04210804, 24.90862058, 28.17855618, 31.90818239] (1.2)

looks good.

Animation

```
norm2:=proc(L) local j; sqrt(add(j^2,j=L)) end:
direction:=proc(L) local j,n:=norm2(L); [seq(j/n,j=L)] end:
makecone:=proc(param,t,length::numeric:=.65,npt::integer:=50,V::list(numeric):=[1,1,1])
local base,sc,i;
    base:=evalc(param);
    sc:=add(V[i]*base[i],i=1..3);
    base:=[seq(length/sc*base[i],i=1..3)];
    plots[polygonplot3d]([[0$3],seq(eval(base,t=2*Pi*i/npt),i=0..npt)],style=surface)
end:
```

The intersection of a sphere with R^3_{>0}:

```
> sph:=plot3d(.6,0..Pi/2,0..Pi/2,coords=spherical,color=
  aquamarine,style=patchnogrid):
```

The directions of the vectors constructed from successive triples of elements of the sequences

```
> WW:=[sph,seq(plottools:-line([0$3],direction(L[i..i+2]),color=
  red,thickness=3),i=1..N-2)]:
> WW2:=[seq(plots[display]([seq(WW[j],j=1..i)]),i=1..nops(WW))]:
> opts:=axes=normal,view=[(-0.25..1)$3],tickmarks=[0$3],
  orientation=[-14,76,-6]:
> plots[display](WW2,insequence=true,opts);
```



```
pict:=plots[display](WW2[-1],opts):
```

Limit matrix and its eigenvalues

```
rec;
```

$$(16\,n+1)\,u(n+3)-(32\,n-2)\,u(n+2)+(20\,n-4)\,u(n+1)-(5\,n-3)\,u(n)=0$$

```
pol:=eval(op(1,rec),u=(k->X^(k-n))):
pol:=collect(pol/lcoeff(pol,X),X,normal):
An:=Transpose(CompanionMatrix(pol,X));
```

(2.2)

$$An := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ \frac{5n-3}{16n+1} & -\frac{4(5n-1)}{16n+1} & \frac{2(16n-1)}{16n+1} \end{bmatrix} \tag{2.2}$$

```
> A:=map(limit,An,n=infinity);
```

$$A := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ \frac{5}{16} & -\frac{5}{4} & 2 \end{bmatrix} \tag{2.3}$$

```
> chpol:=limit(pol,n=infinity);
```

$$chpol := X^3 - 2X^2 + \frac{5}{4}X - \frac{5}{16} \tag{2.4}$$

```
> lambdaf:=[fsolve(chpol,X,complex)];
```

$$lambdaf := [0.4257108730 - 0.3014062877\,I, 0.4257108730 + 0.3014062877\,I, 1.148578254] \tag{2.5}$$

```
> lambdaf:=sort(lambdaf,proc(u,v) abs(u)>=abs(v) end);
```

$$lambdaf := [1.148578254, 0.4257108730 - 0.3014062877\,I, 0.4257108730 + 0.3014062877\,I] \tag{2.6}$$

```
> map(abs,lambdaf);
```

$$[1.148578254, 0.5216085675, 0.5216085675] \tag{2.7}$$

Contracted cone

```
> for i to 3 do V[i]:=Vector([seq(lambdaf[i]^j,j=0..2)]) od;
```

$$V_1 := \begin{bmatrix} 1.0 \\ 1.148578254 \\ 1.319232006 \end{bmatrix}$$

$$V_2 := \begin{bmatrix} 1. + 0.\,I \\ 0.4257108730 - 0.3014062877\,I \\ 0.09038399713 - 0.2566238677\,I \end{bmatrix}$$

$$V_3 := \begin{bmatrix} 1. + 0.\,I \\ 0.4257108730 + 0.3014062877\,I \\ 0.09038399713 + 0.2566238677\,I \end{bmatrix} \tag{3.1}$$

```
> basis:=Matrix([seq(V[i],i=1..3)]);
```

(3.2)

$basis :=$ **(3.2)**

$$\begin{bmatrix} 1.0 & 1. + 0.\,I & 1.\ \text{-}\cdots \\ 1.148578254 & 0.4257108730 - 0.3014062877\,I & 0.4257108730\ \text{-}\cdots \\ 1.319232006 & 0.09038399713 - 0.2566238677\,I & 0.09038399713\ \cdots \end{bmatrix}$$

```
> vv:=Vector([1,c+I*s,c-I*s]);
```

$$vv := \begin{bmatrix} 1 \\ c + I\,s \\ c - I\,s \end{bmatrix} \quad \textbf{(3.3)}$$

```
> border:=evalc(basis.DiagonalMatrix([21/10,1,1]).vv);
```

$border :=$ **(3.4)**

$$\begin{bmatrix} 2.10000000000000 + 0.\,I + 2.\,c \cdots \\ 2.41201433340000 + 0.\,I + 0.851421746000000\,c + 0.602812575400000 \cdots \\ 2.77038721260000 + 0.\,I + 0.180767994260000\,c + 0.513247735400000 \cdots \end{bmatrix}$$

```
> plots[display](pict,makecone(subs(c=cos(t),s=sin(t),border),t,3.
  ,[1,2,3]),orientation=[-14,76,-6]);
```



Contraction index

Symbolic basis (no numerical approximation at all):

```
> Sbasis:=Matrix(3,3,(i,j)->lambda[j]^(i-1));
```

$$Sbasis := \begin{bmatrix} 1 & 1 & 1 \\ \lambda_1 & \lambda_2 & \lambda_3 \\ \lambda_1^2 & \lambda_2^2 & \lambda_3^2 \end{bmatrix} \quad (4.1)$$

Write lambda_2 = alpha + i beta

```
> Sbasis:=subs(lambda[2]=alpha+I*beta,lambda[3]=alpha-I*beta,
  Sbasis);
```

$$Sbasis := \begin{bmatrix} 1 & 1 & 1 \\ \lambda_1 & \alpha + \mathrm{I}\,\beta & \alpha - \mathrm{I}\,\beta \\ \lambda_1^2 & (\alpha + \mathrm{I}\,\beta)^2 & (\alpha - \mathrm{I}\,\beta)^2 \end{bmatrix} \quad (4.2)$$

Extend the first vector so that the cone becomes positive

```
> SBasis:=Sbasis.DiagonalMatrix([21/10,1,1]);
```

$$SBasis := \begin{bmatrix} \frac{21}{10} & 1 & 1 \\ \frac{21\,\lambda_1}{10} & \alpha + \mathrm{I}\,\beta & \alpha - \mathrm{I}\,\beta \\ \frac{21\,\lambda_1^2}{10} & (\alpha + \mathrm{I}\,\beta)^2 & (\alpha - \mathrm{I}\,\beta)^2 \end{bmatrix} \quad (4.3)$$

```
> Sborder:=Sbasis.vv;
```

$$Sborder := \begin{bmatrix} 1 + 2\,c \\ \lambda_1 + (\alpha + \mathrm{I}\,\beta)\,(c + \mathrm{I}\,s) + (\alpha - \mathrm{I}\,\beta)\,(c - \mathrm{I}\,s) \\ \lambda_1^2 + (\alpha + \mathrm{I}\,\beta)^2\,(c + \mathrm{I}\,s) + (\alpha - \mathrm{I}\,\beta)^2\,(c - \mathrm{I}\,s) \end{bmatrix} \quad (4.4)$$

```
> An.Sborder;
```

$$\begin{bmatrix} \lambda_1 + (\alpha \cdots \\ \lambda_1^2 + (\alpha + \cdots \\ \frac{(5\,n - 3)\,(1 + 2\,c)}{16\,n + 1} - \frac{4\,(5\,n - 1)\,(\lambda_1 + (\alpha + \mathrm{I}\,\beta)\,(c + \mathrm{I}\,s) \cdots}{16\,n + 1} \cdots \end{bmatrix} \quad (4.5)$$

```
> Sbasis^(-1).%;
```

$$\left[\begin{array}{c} -\frac{(\alpha + I\beta)(I \ldots}{} \\ -\frac{I\lambda_1 (I\beta - \alpha)(-\alpha + I\beta + \lambda_1)(\lambda_1 + (\alpha \ldots}{2\beta(\alpha^2 - 2\alpha \ldots} \\ \frac{I\lambda_1 (\alpha + I\beta)(\alpha + I\beta - \lambda_1)(\lambda_1 + (\alpha + \ldots}{2\beta(\alpha^2 - 2\alpha\lambda \ldots} \end{array}\right] \quad (4.6)$$

```
> shouldbepositive:=%[1]^2-%[2]*%[3];
```

$$shouldbepositive := \Bigg(\quad (4.7)$$

$$-\frac{(\alpha + I\beta)(I\beta - \alpha)(\lambda_1 + (\alpha + I\beta)(c + Is) + (\alpha - I\beta)(c - Is))}{\alpha^2 - 2\alpha\lambda_1 + \beta^2 + \lambda_1^2}$$

$$-\frac{2\alpha(\lambda_1^2 + (\alpha + I\beta)^2(c + Is) + (\alpha - I\beta)^2(c - Is))}{\alpha^2 - 2\alpha\lambda_1 + \beta^2 + \lambda_1^2}$$

$$+\frac{1}{\alpha^2 - 2\alpha\lambda_1 + \beta^2 + \lambda_1^2}\Bigg(\frac{(5n - 3)(1 + 2c)}{16n + 1}$$

$$-\frac{4(5n - 1)(\lambda_1 + (\alpha + I\beta)(c + Is) + (\alpha - I\beta)(c - Is))}{16n + 1}$$

$$+\frac{2(16n - 1)(\lambda_1^2 + (\alpha + I\beta)^2(c + Is) + (\alpha - I\beta)^2(c - Is))}{16n + 1}\Bigg)\Bigg)^2 - \Bigg($$

$$-\frac{1}{2\beta(\alpha^2 - 2\alpha\lambda_1 + \beta^2 + \lambda_1^2)}(I\lambda_1(I\beta - \alpha)(-\alpha + I\beta + \lambda_1)(\lambda_1 + (\alpha$$

$$+ I\beta)(c + Is) + (\alpha - I\beta)(c - Is)))$$

$$-\frac{I\left(2 I \alpha \beta-\alpha^2+\beta^2+\lambda_1^2\right)\left(\lambda_1^2+(\alpha+I \beta)^2(c+I s)+(\alpha-I \beta)^2(c-I s)\right)}{2 \beta\left(\alpha^2-2 \alpha \lambda_1+\beta^2+\lambda_1^2\right)}$$

$$+\frac{1}{2 \beta\left(\alpha^2-2 \alpha \lambda_1+\beta^2+\lambda_1^2\right)}\Bigg(I\left(-\alpha+I \beta+\lambda_1\right)\Bigg(\frac{(5 n-3)(1+2 c)}{16 n+1}$$

$$-\frac{4(5 n-1)\left(\lambda_1+(\alpha+I \beta)(c+I s)+(\alpha-I \beta)(c-I s)\right)}{16 n+1}$$

$$+\frac{2(16 n-1)\left(\lambda_1^2+(\alpha+I \beta)^2(c+I s)+(\alpha-I \beta)^2(c-I s)\right)}{16 n+1}\Bigg)\Bigg)\Bigg)$$

$$\Bigg(\frac{I \lambda_1(\alpha+I \beta)\left(\alpha+I \beta-\lambda_1\right)\left(\lambda_1+(\alpha+I \beta)(c+I s)+(\alpha-I \beta)(c-I s)\right)}{2 \beta\left(\alpha^2-2 \alpha \lambda_1+\beta^2+\lambda_1^2\right)}$$

$$-\frac{1}{2 \beta\left(\alpha^2-2 \alpha \lambda_1+\beta^2+\lambda_1^2\right)}\Big(I\left(-\beta^2+2 I \alpha \beta+\alpha^2-\lambda_1^2\right)\Big(\lambda_1^2+(\alpha$$

$$+I \beta)^2(c+I s)+(\alpha-I \beta)^2(c-I s)\Big)\Big)$$

$$+\frac{1}{2 \beta\left(\alpha^2-2 \alpha \lambda_1+\beta^2+\lambda_1^2\right)}\Bigg(I\left(\alpha+I \beta-\lambda_1\right)\Bigg(\frac{(5 n-3)(1+2 c)}{16 n+1}$$

$$-\frac{4(5 n-1)\left(\lambda_1+(\alpha+I \beta)(c+I s)+(\alpha-I \beta)(c-I s)\right)}{16 n+1}$$

$$+\frac{2(16 n-1)\left(\lambda_1^2+(\alpha+I \beta)^2(c+I s)+(\alpha-I \beta)^2(c-I s)\right)}{16 n+1}\Bigg)\Bigg)\Bigg)$$

```
> denom(%);
```

$$4 \beta^2\left(\alpha^2-2 \alpha \lambda_1+\beta^2+\lambda_1^2\right)^2(16 n+1)^2 \qquad \textbf{(4.8)}$$

```
> shouldbepositive:=numer(%%);
```

[Length of output exceeds limit of 50000] **(4.9)**

```
> indets(shouldbepositive);
```

$$\left\{\alpha, \beta, c, n, s, \lambda_1\right\} \qquad \textbf{(4.10)}$$

Known constraints on these indeterminates:

```
> evalc(eval(chpol,X=alpha+I*beta));
```

$$\alpha^3-3 \alpha \beta^2-2 \alpha^2+2 \beta^2+\frac{5 \alpha}{4}-\frac{5}{16}+I\left(3 \alpha^2 \beta-\beta^3-4 \alpha \beta+\frac{5}{4} \beta\right) \qquad \textbf{(4.11)}$$

alpha + i beta is a root of the characteristic polynomial

```
> eqs:={coeff(%,I,0),coeff(%,I,1)};
```

(4.12)

$$eqs := \left\{3\,\alpha^2\,\beta - \beta^3 - 4\,\alpha\,\beta + \frac{5}{4}\,\beta,\, \alpha^3 - 3\,\alpha\,\beta^2 - 2\,\alpha^2 + 2\,\beta^2 + \frac{5}{4}\,\alpha - \frac{5}{16}\right\} \tag{4.12}$$

so is lambda_1 and c^2 + s^2 = 1

```
> othereqs:={eval(chpol,X=lambda[1]),c^2+s^2-1}:
> sys:={seq(eq=0,eq=eqs union othereqs)};
```

$$sys := \left\{\lambda_1^3 - 2\,\lambda_1^2 + \frac{5}{4}\,\lambda_1 - \frac{5}{16} = 0,\, 3\,\alpha^2\,\beta - \beta^3 - 4\,\alpha\,\beta + \frac{5}{4}\,\beta = 0,\, \alpha^3 - 3\,\alpha\,\beta^2 - 2\,\alpha^2 + 2\,\beta^2 + \frac{5}{4}\,\alpha - \frac{5}{16} = 0,\, c^2 + s^2 - 1 = 0\right\} \tag{4.13}$$

Simplify using these equality (reduces by a Gröbner basis):

```
> shouldbepositive:=simplify(shouldbepositive,sys);
```

$$\begin{aligned} shouldbepositive := {} & \frac{1}{15872}\Big(\Big(\big((199229440\,n + 12451840)\,c^2 + (-78643200\,n - 5963776)\,c + 134217728\,n^2 - 103809024\,n - 5963776\big)\,\lambda_1^2 \\ & + \big((-356515840\,n - 66519040)\,c^2 + (-15728640\,n - 23789568)\,c + 98566144\,n + 2654208\big)\,\lambda_1 \\ & + (139198464\,n + 38191104)\,c^2 + (50855936\,n + 15237120)\,c - 52428800\,n^2 - 14155776\,n + 4890624\Big)\,\beta^6\Big) \\ & + \frac{1}{31}\Big(221184\Big(\Big(\Big(n - \frac{17}{128}\Big)\,c + \frac{61\,n}{216} - \frac{31}{288}\Big)\,\lambda_1^2 + \Big(\Big(-\frac{443\,n}{288} + \frac{851}{4608}\Big)\,c - \frac{67\,n}{144} + \frac{569}{4608}\Big)\,\lambda_1 + \Big(\frac{251\,n}{432} \\ & - \frac{1111}{13824}\Big)\,c + \frac{163\,n}{864} - \frac{575}{13824}\Big)\,s\,\beta^5\Big) + \frac{1}{15872}\Big(\Big(\big((13303808\,n + 831488)\,c^2 + (-4128768\,n - 3076096)\,c + 19398656\,n^2 - 6750208\,n \\ & - 1742848\big)\,\lambda_1^2 + \big((-16482304\,n - 5638144)\,c^2 + (-11288576\,n + 474112)\,c - 737280\,n + 1451008\big)\,\lambda_1 + (-229376\,n + 676864)\,c^2 + (3096576\,n - 793600)\,c \\ & + 1310720\,n^2 + 1572864\,n - 77824\Big)\,\beta^4\Big) + \frac{1}{31}\Big(23040\Big(\Big(\Big(n - \frac{17}{128}\Big)\,c + \frac{14\,n}{45} - \frac{217}{1920}\Big)\,\lambda_1^2 + \Big(\Big(-\frac{1471\,n}{1440} + \frac{1337}{11520}\Big)\,c - \frac{103\,n}{288} + \frac{973}{11520}\Big)\,\lambda_1 \\ & + \Big(\frac{251\,n}{1920} - \frac{5743}{92160}\Big)\,c + \frac{161\,n}{2880} - \frac{607}{23040}\Big)\,s\,\beta^3\Big) + \frac{1}{15872}\Big(\Big(\big((-3477504\,n + 675456)\,c^2 + (-4919296\,n + 1419008)\,c + 7569408\,n^2 - 1703936\,n \\ & + 422048\big)\,\lambda_1^2 + \big((4971520\,n - 956480)\,c^2 + (6215680\,n - 1855008)\,c - 8888320\,n^2 + 1955840\,n - 589104\big)\,\lambda_1 + (-1297920\,n + 882472)\,c^2 + (\\ & -2318080\,n + 1071608)\,c + 3153920\,n^2 - 889600\,n + 295034\Big)\,\beta^2\Big) \\ & - \frac{1}{31}\Big(1445\,s\Big(\Big(\Big(n - \frac{269}{23120}\Big)\,c + \frac{134\,n}{289} - \frac{3379}{23120}\Big)\,\lambda_1^2 + \Big(\Big(-\frac{568\,n}{289} + \frac{121}{1156}\Big)\,c - \frac{235\,n}{289} + \frac{69}{272}\Big)\,\lambda_1 + \Big(\frac{529\,n}{1156} - \frac{4855}{18496}\Big)\,c + \frac{261\,n}{1156} - \frac{3271}{18496}\Big) \end{aligned} \tag{4.14}$$

$$\beta\Big) + \frac{1}{15872}\Big(\big(\big(-892800\,\alpha^2 + 1097152\,\alpha - 498232\big)\,c^2 + \big(-976128\,\alpha^2 + 1237024\,\alpha - 583544\big)\,c - 223200\,\alpha^2 + 344224\,\alpha - 189782\big)\,\lambda_1^2\Big) + \frac{1}{15872}\Big(\big(\big(1376896\,\alpha^2 - 1235536\,\alpha + 558000\big)\,c^2 + \big(1237024\,\alpha^2 - 1096656\,\alpha + 478640\big)\,c + 274288\,\alpha^2 - 308884\,\alpha + 145390\big)\,\lambda_1\Big) + \frac{\big(-759128\,\alpha^2 + 581560\,\alpha - 215140\big)\,c^2}{15872} + \frac{\big(-583544\,\alpha^2 + 478640\,\alpha - 163060\big)\,c}{15872} - \frac{2009\,\alpha^2}{256} + \frac{1125\,\alpha}{128} - \frac{1735}{512}$$

Positivity of the leading coefficient

The leading coefficient depends only on the limit matrix A and it is positive because the cone is contracted by it:

```
> lco:=lcoeff(shouldbepositive,n);
```

$$lco := \frac{\big(134217728\,\lambda_1^2 - 52428800\big)\,\beta^6}{15872} + \frac{\big(19398656\,\lambda_1^2 + 1310720\big)\,\beta^4}{15872} + \frac{\big(7569408\,\lambda_1^2 - 8888320\,\lambda_1 + 3153920\big)\,\beta^2}{15872} \qquad \textbf{(4.1.1)}$$

```
> lco:=simplify(lco,sys);
```

$$lco := \frac{16\,\beta^2\,\big(16384\,\beta^4\,\lambda_1^2 - 6400\,\beta^4 + 2368\,\beta^2\,\lambda_1^2 + 160\,\beta^2 + 924\,\lambda_1^2 - 1085\,\lambda_1 + 385\big)}{31} \qquad \textbf{(4.1.2)}$$

```
> indets(lco);
```

$$\{\beta, \lambda_1\} \qquad \textbf{(4.1.3)}$$

Proof of positivity by quantifier elimination:

```
> QuantifierElimination[QuantifierEliminate](exists([beta,lambda
  [1]],And(lco<0,subs(X=lambda[1],chpol)=0)));
```

$$\mathit{false} \qquad \textbf{(4.1.4)}$$

Variant:

```
> R:=RegularChains[PolynomialRing]([op(%%)]):
> RegularChains[RealTriangularize]([lco<=0,beta<>0,subs(X=lambda
  [1],chpol)=0],R);
```

$$[\ ] \qquad \textbf{(4.1.5)}$$

The question for general n cannot be answered by QuantifierEliminate or RealTriangularize (killed after several cpu hours).

Approximate cone

Construction of the cone

```
> convert(lambdaf[1],confrac,'K'); K;
```

$$[1, 6, 1, 2, 1, 2, 2, 4, 2, 1, 3, 3]$$

$$\left[1, \frac{7}{6}, \frac{8}{7}, \frac{23}{20}, \frac{31}{27}, \frac{85}{74}, \frac{201}{175}, \frac{889}{774}, \frac{1979}{1723}, \frac{2868}{2497}, \frac{10583}{9214}, \frac{34617}{30139}\right] \quad (5.1.1)$$

```
> lambda_app[1]:=%[2];
```

$$lambda\_app_1 := \frac{7}{6} \quad (5.1.2)$$

```
> evalf(lambdaf[2]);
```

$$0.4257108730 - 0.3014062877\, I \quad (5.1.3)$$

```
> convert(Re(%),confrac,'K'):K;
```

$$\left[0, \frac{1}{2}, \frac{2}{5}, \frac{3}{7}, \frac{20}{47}, \frac{43}{101}, \frac{106}{249}, \frac{149}{350}, \frac{255}{599}, \frac{404}{949}, \frac{1063}{2497}, \frac{7845}{18428}, \frac{8908}{20925}, \frac{16753}{39353}, \frac{25661}{60278}\right] \quad (5.1.4)$$

```
> r2_app:=%[3];
```

$$r2\_app := \frac{2}{5} \quad (5.1.5)$$

```
> convert(Im(lambdaf[2]),confrac,'K'):K;
```

$$\left[-1, 0, -\frac{1}{3}, -\frac{3}{10}, -\frac{19}{63}, -\frac{22}{73}, -\frac{107}{355}, -\frac{343}{1138}, -\frac{793}{2631}, -\frac{5101}{16924}, -\frac{21197}{70327}\right] \quad (5.1.6)$$

```
> i2_app:=%[3];
```

$$i2\_app := -\frac{1}{3} \quad (5.1.7)$$

```
> lambda_app[2]:=r2_app+I*i2_app;lambda_app[3]:=conjugate
  (lambda_app[2]);
```

$$lambda\_app_2 := \frac{2}{5} - \frac{I}{3}$$

$$lambda\_app_3 := \frac{2}{5} + \frac{I}{3} \quad (5.1.8)$$

```
> for i to 3 do Vt[i]:=Vector([seq(lambda_app[i]^j,j=0..2)]) od;
```

$$Vt_1 := \begin{bmatrix} 1 \\ \frac{7}{6} \\ \frac{49}{36} \end{bmatrix}$$

$$Vt_2 := \begin{bmatrix} 1 \\ \frac{2}{5} - \frac{I}{3} \\ \frac{11}{225} - \frac{4\,I}{15} \end{bmatrix}$$

$$Vt_3 := \begin{bmatrix} 1 \\ \frac{2}{5} + \frac{I}{3} \\ \frac{11}{225} + \frac{4\,I}{15} \end{bmatrix} \quad \textbf{(5.1.9)}$$

```
> basis:=Matrix([5/2*Vt[1],Vt[2],Vt[3]]);
```

$$basis := \begin{bmatrix} \frac{5}{2} & 1 & 1 \\ \frac{35}{12} & \frac{2}{5} - \frac{I}{3} & \frac{2}{5} + \frac{I}{3} \\ \frac{245}{72} & \frac{11}{225} - \frac{4\,I}{15} & \frac{11}{225} + \frac{4\,I}{15} \end{bmatrix} \quad \textbf{(5.1.10)}$$

```
> border:=evalc(basis.Vector([1,c+I*s,c-I*s]));
```

$$border := \begin{bmatrix} \frac{5}{2} + 2\,c \\ \frac{35}{12} + \frac{4\,c}{5} + \frac{2\,s}{3} \\ \frac{245}{72} + \frac{22\,c}{225} + \frac{8\,s}{15} \end{bmatrix} \quad \textbf{(5.1.11)}$$

The corresponding cone is positive:

```
> subs(c=-1,s=-1,%);
```

$$\begin{bmatrix} \frac{1}{2} \\ \frac{29}{20} \\ \frac{1663}{600} \end{bmatrix} \quad \textbf{(5.1.12)}$$

Picture

```
> plots[display](pict,makecone(subs(c=cos(t),s=sin(t),border),t,3.
  ,[1,2,3]),orientation=[-14,76,-6]);
```



Check that it is contracted by A

```
> basism1:=basis^(-1);
```

$$basism1 := \begin{bmatrix} \frac{488}{3145} & -\frac{288}{629} & \frac{360}{629} \\ \frac{385}{1258} - \frac{658\,I}{629} & \frac{360}{629} + \frac{3543\,I}{1258} & -\frac{450}{629} - \frac{1035\,I}{629} \\ \frac{385}{1258} + \frac{658\,I}{629} & \frac{360}{629} - \frac{3543\,I}{1258} & -\frac{450}{629} + \frac{1035\,I}{629} \end{bmatrix} \quad \textbf{(5.3.1)}$$

```
> basis^(-1).A.border;
```

$$\begin{bmatrix} \frac{8681}{7548} - \cdots \\ -\left(\frac{1125}{5032} + \frac{5175\,I}{10064}\right)\left(\frac{5}{2} + 2\,c\right) + \left(\frac{755}{629} + \frac{2543\,I}{2516}\right)\left(\frac{3}{1}\cdots\right. \\ \left(-\frac{1125}{5032} + \frac{5175\,I}{10064}\right)\left(\frac{5}{2} + 2\,c\right) + \left(\frac{755}{629} - \frac{2543\,I}{2516}\right)\left(\frac{35}{12}\cdots\right. \end{bmatrix} \quad \textbf{(5.3.2)}$$

```
> evalc(%[1]^2-%[2]*%[3]);
```

$$\left(\frac{8681}{7548} - \frac{367\,c}{15725} - \frac{4\,s}{555}\right)^2 + \left(\frac{2875}{60384} - \frac{100477\,c}{377400} + \frac{467\,s}{1110}\right)\left(-\frac{2875}{60384} + \frac{100477\,c}{377400} - \frac{467\,s}{1110}\right) - \left(\frac{625}{30192} + \frac{5399\,c}{12580} + \frac{38\,s}{111}\right)^2 + I\left(-\left(\frac{625}{30192} + \frac{5399\,c}{12580} + \frac{38\,s}{111}\right)\left(-\frac{2875}{60384} + \frac{100477\,c}{377400} - \frac{467\,s}{1110}\right) - \left(\frac{2875}{60384} - \frac{100477\,c}{377400} + \frac{467\,s}{1110}\right)\left(\frac{625}{30192} + \frac{5399\,c}{12580} + \frac{38\,s}{111}\right)\right) \quad \textbf{(5.3.3)}$$

```
> simplify(%,{c^2+s^2=1});
```

$$\frac{101098663}{98546688} + \frac{376266829\,c^2}{9495384000} + \frac{(-2626445545 - 3959062544\,s)\,c}{56972304000} - \frac{2373209\,s}{33513120} \quad \textbf{(5.3.4)}$$

```
> plot3d(%,c=-1..1,s=-1..1);
```




```
> eval(%%,[c=RealBox(0,1),s=RealBox(0,1)]);
```

⟨RealBox: 1.0259 ± 0.226032⟩ **(5.3.5)**

Variant:

```
> QuantifierElimination[QuantifierEliminate](exists([s,c],And(s^2+
  c^2=1,%%%<=0)));
```

$$false \tag{5.3.6}$$

Contraction index

```
> basis^(-1).An.border:
> evalc(%[1]^2-%[2]*%[3]):
> simplify(%,{c^2+s^2-1});
```

$$\frac{1}{113944608000\,(16\,n+1)^2}\Big(\big(\big(1155891698688\,c^2+(-2027040022528\,s-1344740119040)\,c-2065641113600\,s+29925204248000\big)\,n^2+\big(-3022487522304\,c^2+(-10553949034496\,s-4419254745280)\,c-18899909843200\,s-9637052178800\big)\,n-2191130145792\,c^2+(1884641426432\,s-12482266574240)\,c+4753010670400\,s-15600566720425\Big) \tag{5.4.1}$$

```
> normal(%*(16*n+1)^2);
```

$$\frac{1505067316}{148365375}c^2n^2-\frac{3935530628}{148365375}c^2n-\frac{55941844}{2909125}c^2-\frac{465772064}{26182125}c\,n^2s-\frac{41226363416}{445096125}c\,n\,s+\frac{7361880572}{445096125}c\,s-\frac{1050578218}{89019225}c\,n^2-\frac{373247867}{9623700}n\,c-\frac{78014166089}{712153800}c-\frac{18985672}{1047285}n^2s-\frac{2953110913}{17803845}n\,s+\frac{2970631669}{71215380}s+\frac{101098663}{384948}n^2-\frac{24092630447}{284861520}n-\frac{624022668817}{4557784320} \tag{5.4.2}$$

```
> shouldbepositive:=%:
> st:=time():QuantifierElimination[QuantifierEliminate](exists([s,
  c,n],And(n>=3,s^2+c^2=1,shouldbepositive<0)));time()-st; # about
  30 sec.
```

$$false$$

$$32.640 \tag{5.4.3}$$

Faster using RegularChains:

```
> R:=RegularChains[PolynomialRing]([n,c,s]):
> RegularChains[RealTriangularize]([shouldbepositive<0,n>2,c^2+s^2
  -1=0],R);
```

$$[\ ] \tag{5.4.4}$$

```
> RegularChains[RealTriangularize]([shouldbepositive<0,n=2,c^2+s^2
  -1=0],R);
```

$$[\ ] \tag{5.4.5}$$

```
> RegularChains[RealTriangularize]([shouldbepositive<0,n=1,c^2+s^2
  -1=0],R);
```

$$[regular\_semi\_algebraic\_system,\ regular\_semi\_algebraic\_system,\ regular\_semi\_algebraic\_system,\ regular\_semi\_algebraic\_system] \tag{5.4.6}$$

Interval analysis:

```
> collect(shouldbepositive,n,factor);
```

$$\left(\frac{1505067316}{148365375}c^2-\frac{465772064}{26182125}c\,s-\frac{1050578218}{89019225}c-\frac{18985672}{1047285}s+\frac{101098663}{384948}\right)n^2+\left(-\frac{3935530628}{148365375}c^2-\frac{41226363416}{445096125}c\,s-\frac{373247867}{9623700}c-\frac{2953110913}{17803845}s-\frac{24092630447}{284861520}\right)n-\frac{55941844\,c^2}{2909125}+\frac{7361880572\,c\,s}{445096125}-\frac{78014166089\,c}{712153800}+\frac{2970631669\,s}{71215380}-\frac{624022668817}{4557784320} \tag{5.4.7}$$

```
> eval(%,[c=RealBox(0,1),s=RealBox(0,1)]);
```

⟨RealBox: -136.914 ± 187.03⟩ + (⟨RealBox: 262.629 ± 57.8642⟩) n^2 + (⟨RealBox: -84.5766 ± 323.803⟩) n **(5.4.8)**

```
> collect(%,n,proc(u) Center(u)-Radius(u) end);
```

$$-323.943505749106+204.765206426382\,n^2-408.379604980350\,n \tag{5.4.9}$$

```
> fsolve(%);
```

$$-0.6079306022,\ 2.602310468 \tag{5.4.10}$$

Polyhedral cone

Same basis, different boundary: $|c|+|s|\leq 1$.

More pictures

```
> lambdaf[1];;
```

$$1.148578254 \tag{6.1.1}$$

```
> lambdaf[2];
```

$$0.4257108730-0.3014062877\,\mathrm{I} \tag{6.1.2}$$

```
> plots[display](plots[complexplot](lambdaf,style=point,symbol=
  diagonalcross,symbolsize=20),plots[complexplot](lambdaf[1]*[1,I,
  -1,-I,1]),scaling=constrained,view=[(-1.2..1.2)$2]);
```




```
> opts2:=style=surface,color=cyan:
> border2:=1.4*convert(border/add(border[i],i=1..3),list):
> extr:=seq(eval(border2,[c=dir[1],s=dir[2]]),dir=[[0,1],[1,0],[0,
  -1],[-1,0],[0,1]]);
```

extr := [0.3493207652, 0.5006930968, 0.5499861381], [0.5376700962, 0.4440756721, 0.4182542317], [0.4593510755, 0.4134159679, 0.5272329566], [0.1182099634, 0.5004221784, 0.7813678582], [0.3493207652, 0.5006930968, 0.5499861381] **(6.1.3)**

```
> pp:=plots[polygonplot3d]([[0$3],extr],opts2):
> pp2:=seq(plottools:-line([0$3],i,color="CadetBlue"),i=extr[1..
  -2]):
> plots[display](pict,pp,pp2,opts);
```



```
> basis^(-1).An.border:
> map(evalc@normal,ScalarMultiply(%,16*n+1));
```

$$\left[\begin{array}{l} -\frac{5872}{15725}\,n\,c-\frac{25712}{15725}\cdots \\ \frac{625\,n}{1887}-\frac{1199\,s}{1887}+\frac{7686\,c}{3145}+\frac{21596\,n\,c}{3145}+\frac{608\,n\,s}{111}+\frac{7137}{1509}\cdots \\ \frac{625\,n}{1887}-\frac{1199\,s}{1887}+\frac{7686\,c}{3145}+\frac{21596\,n\,c}{3145}+\frac{608\,n\,s}{111}+\frac{713}{150}\cdots \end{array}\right] \quad (6.1)$$

```
> an:=%[1];
```

$$an := -\frac{5872}{15725}\,n\,c-\frac{25712}{15725}\,c+\frac{34724}{1887}\,n-\frac{4936}{1887}-\frac{64}{555}\,n\,s+\frac{7312}{9435}\,s \quad (6.2)$$

```
> bn:=[coeff(%%[2],I,0),coeff(%%[2],I,1)];
```

$$bn := \left[\frac{625}{1887}\,n-\frac{1199}{1887}\,s+\frac{7686}{3145}\,c+\frac{21596}{3145}\,n\,c+\frac{608}{111}\,n\,s+\frac{71375}{15096},\ -\frac{200954}{47175}\,n\,c+\frac{3736}{555}\,n\,s+\frac{328325}{30192}+\frac{2875}{3774}\,n+\frac{206041}{47175}\,c-\frac{17248}{9435}\,s\right] \quad (6.3)$$

```
> DIRS:=[[0,1],[1,0],[0,-1],[-1,0]];
```

$$DIRS := [[0,1],[1,0],[0,-1],[-1,0]] \quad (6.4)$$

```
> for dir in DIRS do
    val_bn:=eval(bn,[c=dir[1],s=dir[2]]);
    pol[dir]:=eval(an,[c=dir[1],s=dir[2]])-abs(val_bn[1])-abs
  (val_bn[2]);
    n0[dir]:=ceil(fsolve(pol[dir],n));
    pol2:=simplify(pol[dir]) assuming n>=n0[dir];
    print(n0[dir],pol2)
  od:
```

$$4,\ -\frac{452275}{30192}+\frac{18811\,n}{3774}$$

$$2,\ \frac{510919\,n}{47175}-\frac{4310983}{377400}-\frac{\left|-\frac{3834927}{8}+110011\,n\right|}{31450}$$

$$1,\ -\frac{10664}{3145}+\frac{58236\,n}{3145}-\frac{\left|-\frac{26989}{8}+3237\,n\right|}{629}-\frac{\left|-\frac{1917593}{8}+112649\,n\right|}{18870}$$

$$1,\ \frac{681019\,n}{94350}-\frac{3927583}{754800} \quad (6.5)$$

```
> N0:=max(seq(n0[dir],dir=DIRS));
```

$$N0 := 4 \quad (6.6)$$

```
> for dir in DIRS do simplify(pol[dir]) assuming n>=N0 od;
```

$$-\frac{452275}{30192}+\frac{18811\,n}{3774}$$

$$\frac{510919\,n}{47175}-\frac{4310983}{377400}-\frac{\left|-\frac{3834927}{8}+110011\,n\right|}{31450}$$

$$\frac{2215391}{150960} + \frac{139657\,n}{18870}$$
$$\frac{681019\,n}{94350} - \frac{3927583}{754800} \tag{6.7}$$

```
> for dir in DIRS do
    val_bn:=eval(bn,[c=dir[1],s=dir[2]]);
    pol:=eval(an,[c=dir[1],s=dir[2]])-add(signum(lcoeff(co,n))*
  co,co=val_bn);
    fsolve(pol,n);
    n0[dir]:=ceil(max(seq(fsolve(eq,n),eq=[op(val_bn),pol])))
  od;
```

$$val\_bn := \left[\frac{10961\,n}{1887} + \frac{61783}{15096}, \frac{455219}{50320} + \frac{47133\,n}{6290}\right]$$
$$pol := -\frac{452275}{30192} + \frac{18811\,n}{3774}$$
$$3.005389134$$
$$n0_{[0,\,1]} := 4$$
$$val\_bn := \left[\frac{67913\,n}{9435} + \frac{541339}{75480}, -\frac{110011\,n}{31450} + \frac{3834927}{251600}\right]$$
$$pol := \frac{138361\,n}{18870} + \frac{576563}{150960}$$
$$-0.5208864854$$
$$n0_{[1,\,0]} := 5$$
$$val\_bn := \left[-\frac{3237\,n}{629} + \frac{26989}{5032}, \frac{1917593}{150960} - \frac{112649\,n}{18870}\right]$$
$$pol := \frac{2215391}{150960} + \frac{139657\,n}{18870}$$
$$-1.982885749$$
$$n0_{[0,\,-1]} := 3$$
$$val\_bn := \left[-\frac{61663\,n}{9435} + \frac{172411}{75480}, \frac{473783\,n}{94350} + \frac{4911469}{754800}\right]$$
$$pol := \frac{681019\,n}{94350} - \frac{3927583}{754800}$$
$$0.7209018765$$
$$n0_{[-1,\,0]} := 1 \tag{6.8}$$

For n>=5 the polyhedral cone is contracted.

When does the sequence enter the cone?

```
> for i do vv:=basis^(-1).Vector(L[i..i+2]) until vv[1]>abs(coeff
  (vv[2],I,0))+abs(coeff(vv[2],I,1)): i-1;
```

$$3 \tag{6.9}$$